
Lab 4: 
Sample Lab Report
Acme Manufacturing Co. Payroll

Introductory Notes.
t
his program computes payroll for a small sample company of three employees. Employee data is stored as separate rows in a text comma delimited file. The program reads in the weekly pay information for each employee and computes their weekly gross pay including computation of overtime pay (calculated at 50% increased pay for hours worked in excess of 40).

User Interface

The user interface (see Figure 1 below) consists of the following:

· A single command button for calculating payroll.

· A command button to exit the program.

· Picture boxes to display output.

Calculate Payroll Event Handler

After the user clicks the Calculate Payroll button, the program will immediately begin reading file data from the text file AcmePayrollFile.txt. and then will calculate payroll data and store it in the picOut picture box. 

Figure 1: Screen shot of running program

Program Operation Step by Step (Algorithm)

The program operation for the Calculate Payroll Event Handler is given below. It is also included in the comments in the program code

Calculate Payroll Event Handler
1. Declare All constants and variables

2. Clear the table header in the pic box: picHeader

3. Print the table header in the pic box: picHeader

4. Open the input text file for processing: PayrollFile.txt (located in same directory as the program form and code files).

5. Loop through the following code 3 times for each employee.

a. Input a line of three data items from a line of the file to store in local variables for processing (empName, hrlyWage, hrsWorked).

b. Calculate regular pay and overtime pay for two cases:

c. Check IF employee worked more than 40 hours

i. Hours worked > 40:

ii. regPay = 40 * hrlyWage

iii.  otPay = (hrsWorked - 40) * hrlyWage * 1.5 

d. IF employee did NOT work more than 40 hours

i. Hours worked <= 40:

ii. regPay = hrsWorked * hrlyWage

iii. otPay = 0

e. Compute total pay by adding overtime pay to regular pay

f. Print the results for that employee to the picture box

g. Loop back and do the next employee.

6. After looping is complete for reading, processing, and printing the employee data to the picture box, close the input file.

7. Exit the event handler.

Main Form and Event Handler Code

Program Code for the ‘Calculate Payroll’ event handler is below. 

Private Sub cmdCalculate_Click()

    'Program Version and Author

        'Program Title: payroll1

        'File Names: payroll1.vbp , payroll1.frm

        'Date: January 27, 2003

        'Author: J.Pejsa

        'Class: BSAD 179 Winter 2003 - Instructor: J. Pejsa

        'Assignment: This program documents Lab Assignment 1

        'Due Date: Jan 28, 2003

        'Version 1.0

    'Declare all my variables

    Dim empName As String, hrlyWage As Single

    Dim hrsWorked As Single

    Dim regPay As Single, otPay As Single

    Dim totPay As Single

    Dim loopCounter As Integer

    'Print the header of the output table in the pic box

    picHeader.Cls

    picHeader.Print "Name", "Wage", "Hours", "Reg Pay",

    picHeader.Print "OT Pay", "Total Pay"

    'Also clear the main big output picBox

    picOut.Cls

    'Open the text file called 'AcmePayrollFile.txt'

    '   (it is in same directory as this program)

    Open App.Path & "\AcmePayrollFile.txt" For Input As #1

    'Read 3 employee records from file for processing

    For loopCounter = 1 To 3

    'For each of the 3 employee records, we do the following:    

          'Read in the first record of employee data from file

          '   into 3 local vars: empName, hrlyWage, hrsWorked

          Input #1, empName, hrlyWage, hrsWorked

          '--------------------------------------------------

          'Compute the regular and OT pay for this employee

          '    store into local vars regPay and otPay

          '    use an IF statement to check for overtime pay

          If hrsWorked > 40 Then

          'we know with certainty employee worked over 40 hrs              

              regPay = 40 * hrlyWage

              otPay = (hrsWorked - 40) * hrlyWage * 1.5

              '    compute overtime pay at 1.5 times reg pay

          Else

          'we know with certainty employee did NOT work over 40 hrs          

              regPay = hrsWorked * hrlyWage

              otPay = 0

          End If

          '--------------------------------------------------                          

          'Add the regPay and otPay together to get totPay

          '    for that employee

          totPay = regPay + otPay          

          'Print out the employee data to the pic box

          '   Payroll total for that employee in var totPay

          picOut.Print empName, FormatCurrency(hrlyWage),

          picOut.Print hrsWorked,

          picOut.Print FormatCurrency(regPay),

          picOut.Print FormatCurrency(otPay),

          picOut.Print FormatCurrency(totPay)        

    Next loopCounter

    'End of main loop for reading and processing file records        

    'Close the text file.

    Close #1

End Sub

Program Code for the ‘Exit Program’ event handler is below. 

Private Sub cmdExitProgram_Click()

  'Terminate the program

  End  

End Sub 

Input Data

Below is a screen shot of the input data file contents used by this sample program. Note that only 3 employees were used for this sample program.

Col 1: Name of Employee, Col 2: Employee Wage, Col 3: Hours worked.  

Figure 2: Sample Employee Payroll Data in file: AcmePayrollF 

Program Limitations - Improvement Suggestions 

1. Program will only work for three employees.

2. Generalize program to work with any input data.

3. User should be able to scroll through output data

4. Results should be able to be printed as a report.

5. Computation of taxes and payroll deductions should be considered.

